Kalman example

Generic study (without defining parameters)

Let denote K = Q (R1, R2, L, C) and let us define the Ore algebra A = K(t)⟨∂⟩, where ∂ is differential operator defined by $\partial f(t) = f'(t)$.

```
A = OreAlgebra[Der[t]]
```

$\mathbb{K}(t)[D_t; 1, D_t]$

We consider the system defined by the matrix $R \in A^{2\times 3}$ given by

```
MatrixForm[R = ToOrePolynomial[{{Der[t] + 1/(R1 C), 0, -1/(R1 C)},
    {0, Der[t] + R2/L, -1/L}}, A]]
```

$$\begin{pmatrix} D_t + \frac{1}{C\,R1} & 0 & -\frac{1}{C\,R1} \\ 0 & D_t + \frac{R2}{L} & -\frac{1}{L} \end{pmatrix}$$

that is the system defined by:

```
Thread[ApplyMatrix[R, {x1[t], x2[t], u[t]}] == 0]
```

$$\left\{\frac{-u[t] + x_1[t] + C\,R1\,x_1'[t]}{C\,R1} == 0,\ \frac{-u[t] + R2\,x_2[t] + L\,x_2'[t]}{L} == 0\right\}$$

and the left A-module, finitely presented by R, is M = $A^{1\times 3} / A^{1\times 2} R$.

Let us compute the adjoint of the system

```
MatrixForm[Radj = Involution[R, A]]
```

$$\begin{pmatrix} -D_t + \frac{1}{C\,R1} & 0 \\ 0 & -D_t + \frac{R2}{L} \\ -\frac{1}{C\,R1} & -\frac{1}{L} \end{pmatrix}$$

```
MatrixForm /@ ({Ann, Rp, Q} = Exti[Radj, A, 1])
```

$$\left\{\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & -L\,D_t - R2 & 1 \\ -C\,R1\,D_t - 1 & 0 & 1 \end{pmatrix}, \begin{pmatrix} L\,D_t + R2 \\ C\,R1\,D_t + 1 \\ C\,L\,R1\,D_t^2 + (L + C\,R1\,R2)\,D_t + R2 \end{pmatrix}\right\}$$

Since the first matrix is identity matrix, we know there is generically no autonomous element (because M is generically torsion free), where the term generically means for almost all values of the parameter R1, R2, C and L. This will be explained below.

Let us compute a flat output ξ(t) of the system

```
MatrixForm[T = LeftInverse[Q, A]]
```

$$\begin{pmatrix} \frac{C\,R1}{-L + C\,R1\,R2} & \frac{L}{L - C\,R1\,R2} & 0 \end{pmatrix}$$

```
ξ = ApplyMatrix[T, {x1[t], x2[t], u[t]}]
```

$$\left\{\frac{-C\,R1\,x_1[t]+L\,x_2[t]}{L-C\,R1\,R2}\right\}$$

Let us now study the particular case when L-C R1 R2 =0. For this let us define a new Ore algebra B, in which we add the relation L-C R1 R2 =0 between parameters.

```
B = OreAlgebra[Der[t], CoefficientNormal -> (Expand[# //. L -> C R1 R2] &)]
```

$\mathbb{K}(t)[D_t;\,1,\,D_t]$

The new matrix of the system is then

```
MatrixForm[S = ToOrePolynomial[{{Der[t] + 1/(R1 C), 0, -1/(R1 C)},
    {0, Der[t] + R2/L, -1/L}}, B]]
```

$$\begin{pmatrix} D_t+\frac{1}{C\,R1} & 0 & -\frac{1}{C\,R1} \\ 0 & D_t+\frac{1}{C\,R1} & -\frac{1}{C\,R1\,R2} \end{pmatrix}$$

the new system is defined by

```
Thread[ApplyMatrix[S, {x1[t], x2[t], u[t]}] == 0]
```

$$\left\{\frac{-u[t]+x_1[t]+C\,R1\,x_1'[t]}{C\,R1}=0,\ \frac{-u[t]+R2\,(x_2[t]+C\,R1\,x_2'[t])}{C\,R1\,R2}=0\right\}$$

and the new left B-module, finitely presented by S, is N = $B^{1\times 3}\big/B^{1\times 2}\,S$.

Let us compute the adjoint of the system

```
MatrixForm[Sadj = Involution[S, B]]
```

$$\begin{pmatrix} -D_t+\frac{1}{C\,R1} & 0 \\ 0 & -D_t+\frac{1}{C\,R1} \\ -\frac{1}{C\,R1} & -\frac{1}{C\,R1\,R2} \end{pmatrix}$$

```
MatrixForm /@ ({Ann, Sp, Q} = Exti[Sadj, B, 1])
```

$$\left\{\begin{pmatrix} -C\,R1\,D_t-1 & 0 \\ 0 & 1 \end{pmatrix},\ \begin{pmatrix} -1 & R2 & 0 \\ 0 & C\,R1\,R2\,D_t+R2 & -1 \end{pmatrix},\ \begin{pmatrix} R2 \\ 1 \\ C\,R1\,R2\,D_t+R2 \end{pmatrix}\right\}$$

Since the first matrix is not identity, the system is not controllable (N is not torsion free) and we have one autonomous element, defined by the first row of Sp, namely θ, defined by

```
ApplyMatrix[{Sp[[1]]}, {x1[t], x2[t], u[t]}]
```

$\{-x_1[t]+R2\,x_2[t]\}$

Moreover, θ satisfies the equation

```
Thread[ApplyMatrix[{{Ann[[1, 1]]}}, {θ[t]}] == 0]
```

$\{-\theta[t]-C\,R1\,\theta'[t]=0\}$

Finally, this result can be obtained directly by means of the command AutonomousElements

```
AutonomousElements[S, {x1[t], x2[t], u[t]}, θ, B]
```

$\{\{\theta[1][t] \to -x_1[t] + R2\, x_2[t],\ \theta[2][t] \to -u[t] + R2\, (x_2[t] + C\, R1\, x_2'[t])\},$
$\{-\theta[1][t] - C\, R1\, \theta[1]'[t] == 0,\ \theta[2][t] == 0\}\}$